@echo off

:: 环境标识
SET ENV=PRE
:: 当前工作目录
SET WORK_PATH=%cd%
call lib\tryMkdir "%WORK_PATH%\build\"

:: 作用是在循环体重使用感叹号标记变量，可以获取变量的最新值
setlocal EnableDelayedExpansion

:: 遍历 hosts.conf 配置文件每行文本
for /f  "skip=1 tokens=*" %%H in (hosts.conf) do (
	:: %%H 变量代表的是其中一行文本数据，如果是空行则跳过不处理
	if "%%H" NEQ "" (
		:: 进一步遍历该行中的每个参数，以井号分割
		for /f "tokens=1,2,3,4,5,6,7 delims=#" %%a in ("%%H") do (
			:: 该行中的每个参数将会分别赋值费 %%a、%%b、%%c...
			if "%%a" == "%ENV%" (
				:: 判断待发布的jar包是否存在，存在则执行发布，否则跳过
				if exist %WORK_PATH%\build\%%g (
					:: 获取一个随机数字串（用于当临时文件的名字使用）
					SET RANDOM_NUM=!RANDOM!!RANDOM!!RANDOM!!RANDOM!!RANDOM!!RANDOM!
					:: 调用部署命令，每台机器对应一个窗口，并行发布
				  	start lib\deploy %%a %%b %%c %%d "%%e" %%f %%g %WORK_PATH% !RANDOM_NUM!
				)
			)
		)
	)
)
endlocal




